# Chapter 3

# Taxes, Tax Reform, Public Goods, and Steady-State Models Introduces Auxiliary Variables, Constraints, and Rationing

In this chapter, we will begin with a simple problem in which there are taxes in the initial benchmark data, to illustrate how to calibrate a model with existing taxes. Then we will consider labor taxes with endogenous labor supply. These models will make the point that some taxes are less distortionary than others. However, replacing one tax with another rarely yields the same revenue in general equilibrium. Thus our third example will show how to model tax reforms that yield the government the same amount of revenue, referred to as "equal-yield tax reform". In doing so, we will introduce a major feature of MPS/GE, the auxiliary variable and the constraint equation. The new tax rate will be an endogenous tax, whose value is set in general equilibrium by the revenue constraint.

Next, we will introduce a public good, and show how that is modeled. Then we will set the tax rate to finance this public good endogenously according to the Samuelson rule for pure public goods. This will also involve the use of an endogenous tax rate, an auxiliary variable, and a constraint equation. But it will also involve the introduction of another new feature, the rationing constraint.

Finally, we will end this chapter with two other extensions that are valuable for policy analysis. First, we will consider taxes and classical unemployment. Second, we will consider an endogenous capital stock that adjusts according to a steady-state rule. Comparative statics experiments are then in fact comparative steady-state experiments, showing the general-equilibrium change in all variables after the capital stock has adjusted to its new general-equilibrium level.

The models in this chapter are as follows:

M31.GMS	introduces taxes in the benchmark
M32.GMS	labor supply with a benchmark labor tax
M33.GMS	introduces equal yield tax reform
M34.GMS	introduces a public good
M35.GMS	public good provision using the samuelson rule
M36.GMS	taxes and classical unemployment

M37.GMS steady-state capital stock

## Model M31 introduces taxes in the benchmark

Our first model is similar to the models of the previous chapter, but a positive tax and tax revenue are present in the benchmark data. The first task of the modeler is again to construct a micro-consistent data set, and the format we introduced in chapter 1 (dubbed the micro-consistency matrix) is a nice way to do this and at the same time set up the data for building the model. Remember that entries into this matrix are values. Row sums equal to zero indicate market clearing and column sums equal to zero indicate product exhaustion (zero profits) or consumer income balance.

Under this convention, each tax should be added as a row to the matrix. Taxes are negative entries in a column indicating payments by a sector. There is a corresponding positive entry somewhere. In the present case, the tax is simply redistributed lump sum to the consumer, so the consumer gets a positive entry of the tax revenue. A zero row sum for the tax indicates that all tax receipts must be paid to someone. Here is our matrix:

	Prod	Consumers		
Markets	x	Y	W	CONS
PX PY PW PL PK TAX	-20 -60 -20	100 -60 -40 0	-100 -100 200	-200 80 100 20

The data indicate that the X sector receives 100 units of revenue, of which 20 is paid in taxes. This 20 is received as part of the consumer's income. Note that these data do not indicate what type of tax is in place. It could be a tax on X output, on all the inputs, or on just one input. We are going to interpret this as a tax on the labor input into sector X.

A crucial task for the modeler is to keep track of what prices firms and consumers face. It is (generally) not possible to calibrate a benchmark equilibrium with all prices equal to one. For example, if a production input is taxed, then if its consumer price (price received by the consumer) is chosen to be equal to one, then producer price (price paid by the producer) is specified as (1 + t). On the other hand, if the producer price is unity, the consumer price is 1/(1+t). We will discuss output taxes shortly.

Given that we interpret the above data as a tax on the labor input into the X sector, the data tell us that the tax rate is 100%. The amount paid by the X sector to labor (20) is equal to

the tax revenue (20). Thus if we set the consumer price of labor to 1 (also the price to the Y sector), then the price of labor to the X sector must be 2.

When the benchmark price of a production input or output is not equal to unity, it is necessary to add a reference price field to the production block. The relative prices of inputs fix the marginal rate of substitution (on inputs, marginal rate of transformation on outputs). Benchmark reference prices and reference quantities are needed in order to correctly fit (calibrate) the technology to the data. If these fields are not correctly specified, then a run of the model will not reproduce the benchmark data as an equilibrium.

Here is the correct specification of the X production block (recall that default prices equal 1 so only prices not equal to 1 must be specified).

```
$PROD:X s:1
O:PX Q:100
I:PL Q: 20 P:2 A:CONS T:TLX
I:PK Q: 60
```

where TLX (for tax on labor in X) is declared as a parameter elsewhere in the file, in order to permit changes in TLX as counterfactual experiments.

There are two common mistakes that you need to avoid. First is to leave out the price field thinking that the assignment of the tax will do that for you. That is incorrect, the P:2 must appear in the block. The tax itself is not used in the calibration, the Q and P fields are.

Second, the P field must *not* be specified as P:(1 + TLX). While this would result in the correct initial calibration, any subsequent change in TLX in a counterfactual would result in a recalibration of the technology. Assuming that the modeler wants to just change the tax, not the technology, the P field must have a constant in it. Of course, this constant could itself be a parameter specified somewhere else in the file (e.g., P:(1 + TLX0)), but it must be constant and distinct from the changing tax rate.

In some instances, the modeler may of course want to change the technology itself as a counterfactual. That is fine, but be clear on what it is you want to change.

Finally, we are going to allow for alternative taxes in the model, including a tax on capital inputs into X (TKX) and a tax on X output (TX). So the production block in the model is written as follows, where TKX and TX are set to zero initially.

```
$PROD:X s:1
O:PX Q:100 A:CONS T:TX
I:PL Q: 20 P:2 A:CONS T:TLX
I:PK O: 60 A:CONS T:TKX
```

In the first counterfactual, eliminate taxes on X sector inputs and replace with a single tax on X sector output. We make a guess that the 100% tax on labor might generate revenue roughly equivalent to a tax of 25% on both inputs: 20 units of revenue is 100% of the benchmark labor input, or 25% of total benchmark inputs of labor and capital (20 + 60 = 80). Of course, they will not in fact generate the same revenue as other variables adjust in general equilibrium. Further discussion of this issue is postponed until model M33.

Now consider a tax on X sector output. It is important to understand that the output tax rate will be different from the corresponding tax rate on all inputs, because the tax base is different in the two cases. Let mc denote the marginal cost of production (or producer price) and p denote the price charged to the consumer. This is how MPS/GE interprets input (ti) versus output (to) taxes.

Tax on all inputs: p = (1 + ti)mcTax on the output: p(1 - to) = mc

Note mc is the tax base for the input tax, and p is the tax base for the output tax. These are not the same. The output tax that is equivalent to the tax on all inputs is found by:

$$(1+ti)=\frac{1}{1-to}$$

If ti = TLX = TKX = 0.25 as we have assumed in our first counterfactual, then the equivalent output tax is given by to = TX = 0.20. Thus in our second counterfactual, we set the input taxes to zero and the output tax to 0.20.

One more equivalence can be demonstrated with this small model. The final counterfactual demonstrates that a 20% tax on the output of X is the same as a 25% subsidy to the production of Y. Let t be the tax on X and s the subsidy to Y. Formally, we have

$$\frac{p_x(1-t)}{p_y} = \frac{p_x}{p_y(1+s)} = \frac{mc_x}{mc_y} \qquad if \quad t = 0.20, \quad s = 0.25$$

Absolute prices may differ depending on the choice of the numeraire, but all quantities and welfare are the same.

Here is the program.

\$TITLE Model M31.GMS: Closed 2x2 Economy - Calibrate to an Existing Tax

### \$ONTEXT

	Prod	luction S	Sectors	Consumers
Markets	x	Y	W	CONS
PX PY PW PL PK TAX	100     -20   -60   -20	100 -60 -40 0	-100 -100 200	-200 80 100 20

### \$OFFTEXT

#### PARAMETERS

TX	Proportional output tax on sector X,
TY	Proportional output tax on sector Y,
TLX	Ad-valorem tax on labor inputs to X,
TKX	$Ad ext{-valorem}$ tax on capital inputs to $X$ ;

### \$ONTEXT

### \$MODEL:M31

## \$SECTORS:

X ! Activity level for sector X
Y ! Activity level for sector Y

W ! Activity level for sector W (Hicksian welfare index)

# \$COMMODITIES:

PX ! Price index for commodity X PY ! Price index for commodity Y

PL ! Price index for primary factor L (net of tax)

PK ! Price index for primary factor K

PW ! Price index for welfare (expenditure function)

## \$CONSUMERS:

CONS ! Income level for consumer CONS

### \$PROD:X s:1

O:PX Q:100 A:CONS T:TX I:PL Q: 20 P:2 A:CONS T:TLX

```
I:PK
              Q: 60
                             A:CONS T:TKX
$PROD:Y s:1
        O:PY
                Q:100
                             A:CONS T:TY
        I:PL
                Q: 60
        I:PK
                Q: 40
$PROD:W s:1
        O:PW
                Q:200
        I:PX
                Q:100
        I:PY
                Q:100
$DEMAND: CONS
        D:PW
                Q:200
                Q: 80
        E:PL
        E:PK
                Q:100
$OFFTEXT
$SYSINCLUDE mpsgeset M31
PW.FX = 1;
TX = 0;
TY = 0;
TLX = 1;
TKX = 0;
$INCLUDE M31.GEN
SOLVE M31 USING MCP;
        In the first counterfactual, we replace the tax on labor inputs
        by a uniform tax on both factors:
TLX = 0.25;
TKX = 0.25;
TX = 0;
TY = 0;
$INCLUDE M31.GEN
SOLVE M31 USING MCP;
        Now demonstrate that a 25% tax on all inputs is equivalent to a
        20% tax on the output (or all outputs if more than one)
TLX = 0;
TKX = 0;
TX = 0.2;
TY = 0;
```

## Exercises:

- (1) Verify that the results from all the counterfactuals accord with our theoretical intuition
- (2) Set the price field for PL in the PROD:X block in the benchmark equal to one, deliberately creating an error. Run the model and verify that the first solution does not reproduce the benchmark.
- (3) Deliberately create a second error. Set the price field equal to P:(1 + TLX). Run the model, and see that this replicates the benchmark, but gives different answers to the counterfactuals than with P:2.

# Model 32 closed economy 2x2 with labor supply and labor tax

This model is an extension of the previous model and also extends our earlier model with endogenous labor supply (M26) to a case with taxes in the benchmark. Since the extension here is fairly straightforward, we will take the opportunity to introduce two useful features. One is to first run the model without allowing it to iterate. This allows the modeler to check if the initial values of the model are an equilibrium; that is, is it calibrated correctly? If it is not calibrated correctly, it will indicate what activities or markets are out of balance which is very useful in correcting the calibration. Second, when the modeler wants to loop over a set of parameter values repeatedly solving the model, there is a simple procedure for doing this.

Here is the initial data for the model.

	C	Consumers				
Markets	A	В	W	TL	TK	CONS
PX	120		-120			
PY		120	-120			
PW			340			-340
PLS	-48	-72		120		
PKS	-72	-48			120	
${ t PL}$			-100	-100		200
PK					-100	100
TAX	 			-20	-20	40

There are supply activities for labor (TL) and capital (TK). Labor can also be used for leisure and so the activity level for labor supply will vary. Capital has no alternative use so it will always be completely supplied to the market. Still, it can be convenient to specify a supply activity, since the tax on capital supply need only be specified once and there will be two prices, one the consumer price and one the producer price (user cost) of capital.

As in the previous model, we have to make a choice of units for prices. Our choice will be that the consumer prices (prices received by the consumer) for labor and capital will be set to one. The data matrix indicates that there is a 20% tax on each factor in the benchmark, so the producer prices (user costs) of labor and capital will be PLS = PKS = 1.2.

We can also choose how to interpret the X and Y values, but there is only a single price for both producers and consumers, so we will interpret these as 120 units at a price of 1 for each. From here on, things are rather straightforward, so let us introduce a couple of other useful features.

First, in complicated models with lots of sectors and taxes, benchmarking is a difficult task and it is often not possible to calibrate the model with all prices and activity levels equal to

one. One useful trick for checking the calibration and noting which sectors or markets are out of balance is to not allow the model to iterate initially. After the MPS/GE model itself, you will see the notation:

```
PLS.L =1.2; PKS.L =1.2;
M32.ITERLIM = 0;
$INCLUDE M32.GEN
SOLVE M32 USING MCP;
M32.ITERLIM = 2000;
```

First, we set initial values for any activities or prices that we know do not take on the default values of 1. This is done with the .L suffix on the variable that we used once before. Note that this does not fix the value of the variable (that is the .FX suffix), it just sets its initial value.

Next we use a command which tells GAMS that it cannot iterate in attempting to solve the model (syntax is <model name>.ITERLIM = 0;). Then we use the INCLUDE and SOLVE statements. Then, in preparation for the next run we set the iteration limit back to a high value.

Suppose that we had set the initial value of PLS.L = 1.0 instead of 1.2. This is what you will see if you look at the listing file (M32.LST).

		LOWER	LEVEL	UPPER	MARGINAL
VA	R X	•	1.000	+INF	-8.440
VA	AR Y	•	1.000	+INF	-12.435
VA	AR W	•	1.000	+INF	•
VA	R TL	•	1.000	+INF	20.000
VA	R TK	•	1.000	+INF	•
VA	R PX	•	1.000	+INF	
VA	R PY	•	1.000	+INF	•
VA	R PL	•	1.000	+INF	•
VA	R PK	•	1.000	+INF	•
VA	R PLS	•	1.000	+INF	-9.163
VA	R PKS	•	1.200	+INF	8.365
VA	R PW	1.000	1.000	1.000	EPS
VA	R CONS	•	340.000	+INF	•

The model has not solved. Recall from chapter 1 that GAMS writes inequalities in the greater-than-or-equal-to format. The MARGINAL column of the listing file gives the degree of imbalance in an inequality, left-hand side minus right-hand side. A positive number is ok if the associated variable is zero, as in a cost equation (marginal cost minus price is positive if

associated with a slack activity). A negative value of a marginal cannot be an equilibrium; for an activity it indicates positive profits and for a market it indicates demand exceeds supply. In our incorrect calibration in which we give the producer price of labor too low a value, we see that there are positive profits for X, Y and negative profits for labor supply. There is an excess demand for labor and an excess supply for capital.

Most calibration errors are in the MPS/GE file itself, and not just in setting the initial values of the variables. You could work with this file as an exercise, deliberately introducing errors (such as in the price fields) and see what happens. In any case, the iterlim = 0 statement is very useful in helping you identify where the errors are.

The other useful feature we introduce in this model is the use of the LOOP statement to simplify the repeated solving of the model over a series of parameter values. First, we use a set statement to indicate a set of values to be looped over. We cannot go through all of the possible ways to do this in GAMS, here we just use a very simple formulation in which there are five values in the set, denoted just 1-5 (they could be called S1-S5, etc.). Two parameters are declared as vectors, WELFARE(S), and LABSUP(S) (for labor supply).

Then the loop statement sets the taxes at different values over the values of the set.

```
LOOP(S,

TXL = 0.25 - 0.05*ORD(S);

TXK = 0.15 + 0.05*ORD(S);

$INCLUDE M32.GEN

SOLVE M32 USING MCP;

WELFARE(S) = W.L;

LABSUP(S) = TL.L;

);

DISPLAY WELFARE, LABSUP;
```

ORD(S) denotes the ordinal value of a member of a set. S is an indicator and is not treated as a number in GAMS, so 0.05\*S won't work. ORD(S) is treated as a number, so this is how the set index is translated into a number. Note from the tax assignment statement that when S=1, the initial values of both taxes are 0.20, our benchmark values. At S=5, the values are TXL=0, and TXK=0.40.

The model is repeatedly solved within the loop, and after each solve statement the value of the parameters WELFARE and LABSUP are assigned values. The loop is closed with "); "After the loop is closed we ask GAMS to display the parameters at the end of the listing file. Note the set index for the parameters is not used in the display statement, GAMS knows what it is.

\$TITLE Model M32: Closed 2x2 Economy -- income taxes and labor supply \$ONTEXT

	Prod	Sectors		(	Consumers	
Markets	x	Y	W	TL	TK	CONS
PX	120		-120			
PY	İ	120	-120			
PW			340			-340
PLS	-48	-72		120		
PKS	-72	-48			120	
${ t PL}$			-100	-100		200
PK					-100	100
TAX				-20	-20	40

#### \$OFFTEXT

- \* Declare parameters to be used in setting up counter-factual
- \* equilibria:

SETS S /1\*5/;

#### PARAMETERS

TXL Labor income tax rate,
TXK Capital income tax rate,

WELFARE(S) Welfare,

LABSUP(S) Labor supply;

#### \$ONTEXT

\$MODEL:M32

## \$SECTORS:

X ! Activity level for sector X

Y ! Activity level for sector Y

W ! Activity level for sector W (Hicksian welfare index)

TL ! Supply activity for L TK ! Supply activity for K

### \$COMMODITIES:

PX ! Price index for commodity X PY ! Price index for commodity Y

PL ! Price index for primary factor L (net of tax)
PK ! Price index for primary factor K (net of tax)

```
PLS
                ! Price index for primary factor L (gross of tax)
       PKS
                ! Price index for primary factor K (gross of tax)
       PW
                ! Price index for welfare (expenditure function)
$CONSUMERS:
               ! Income level for consumer CONS
$PROD:X s:1
       O:PX
              Q:120
        I:PLS Q: 40
                    P:1.2
        I:PKS Q: 60 P:1.2
$PROD:Y s:1
       O:PY
              Q:120
        I:PLS Q: 60
                     P:1.2
       I:PKS Q: 40
                     P:1.2
$PROD:TL
       O:PLS Q:100
                     P:1.2
       I:PL
              Q:100
                     P:1
                            A:CONS T:TXL
$PROD:TK
       O:PKS Q:100 P:1.2
       I:PK
              Q:100 P:1
                            A:CONS T:TXK
$PROD:W s:0.7
              a:1
       O:PW
              Q:340
        I:PX
              Q:120
                     a:
        I:PY
             Q:120
                     a:
       I:PL
             Q:100
$DEMAND: CONS
       D:PW
             Q:340
       E:PL
              Q:200
       E:PK
              Q:100
$OFFTEXT
$SYSINCLUDE mpsgeset M32
       Benchmark replication:
TXL = 0.2;
TXK = 0.2;
PW.FX = 1;
PX.L =1.; PY.L =1.; PLS.L =1.2; PKS.L =1.2;
```

## Exercises:

- (1) As suggested above in M31, deliberately create the same calibration errors in the MPS/GE file and see how this affects the attempt to reproduce the benchmark data using the iterlim = 0 statement.
- (2) Change the elasticity of substitution between leisure and goods from 0.7 to 0.5. See if you can guess how this should effect the results from the counter-factuals. Should the shift to capital taxation increase welfare faster or not?
- (3) In order to get more comfortable with setting producer versus consumer prices, reinterpret the bencmark data so that the producer prices (user costs) of capital and labor are equal to one. So, for example, the value of labor in the X sector of 48 is 48 units at a price of 1. Rewrite the MPS/GE file using this convention.

## Model 33 introduces equal yield tax reform

This model is a follow-up on model M32 where we considered some income tax reform experiments. Like that model, here we apply taxes through activities which transform (supply) household owned factors into production inputs. The difference here is that we set up a model in which we can do differential tax policy analysis holding the level of government revenue constant. In order to keep things simple, we continue to rebate tax revenue in lump-sum fashion.

This model introduces a fourth (and final) class of MPSGE variables (in addition to activity levels, commodity prices and income levels). The new entity is called an "auxiliary variable". In this model, we use an auxiliary variable to endogenously alter the tax rate in order to maintain an equal yield.

In the present case, we will hold the labor tax rate exogenous, but change its value, in each case solving for the value of the capital tax that yields the same value of revenue as the original tax. TXK now become a *variable*, not a *parameter*. In the initial MPS/GE statements specifying the model, we declare a list of "auxiliary variables" (note the spelling of auxiliary, use a double "l" and the model will crash), in this case a single variable TXK.

```
$AUXILIARY:
TXK ! Endogenous capital tax from equal yield constraint.
```

This variable appears in the production block for capital supply as follows.

```
$PROD:TK
O:PKS Q:100 P:1.2
I:PK Q:100 A:CONS N:TXK
```

N: is a new field in a production block, standing for eNdogenous tax rate followed by the auxiliary variable. This is read in words as "assign to consumer CONS the revenue from an endogenous tax rate TXK". TXK is in turn associated with a constraint equation, which is written in GAMS syntax but inside the MPS/GE block. Here is how we write it in this model.

```
$CONSTRAINT:TXK

TXL*PL*TL*100 + TXK*PK*TK*100 =E= 40*(PX + PY)/2;
```

The left-hand side is tax revenue from the two taxes, one an exogenous parameter (TXL) and the other an endogenous variable (TXK). Each term is (tax rate) x (factor price) x (activity level for factor supply) x (the reference quantity supplied at an activity level equal to one).

The right-hand side of the constraint specifies the target revenue. With prices changing in general equilibrium, the modeler has to think carefully about what is meant by "constant" revenue: that is, constant in terms of what? Here we assume that the government wants the taxes to yield an amount equal to the cost of purchasing given and equal amounts of X and Y, so for example the initial tax revenue of 40 is spend (at prices for X and Y equal to one) on 20 X and 20 Y. The right-hand side of the constraint equation will continue to allow the purchase of

X = Y = 20 as the prices of the goods change in general equilibrium. Of course, the government is not actually buying anything in this simple model, it is just redistributing the revenue back to the consumer. But the point is that the modeler must specify what the revenue target is in real terms, and then allow the nominal value of that to adjust with prices.

For historical reasons, the default values of auxiliary variables in MPS/GE are zero, not one. So the modeler should set the initial values of these variables for the replication check even if they should happen to be one. In our case, the initial value of TXK = 0.20, so we set this along with the values of PLS and PKS which are equal to 1.2 initially, along with the initial value of the parameter TXL (the latter is a parameter and so does not use the '.L' syntax).

```
TXL = 0.20;

TXK.L = 0.20;

PLS.L = 1.2;

PKS.L = 1.2;
```

After the replication check, we loop over values of TXL, and each solve statement finds the new value of TXK as one variable in the new general-equilibrium solution.

In each iteration, we store the values of key variables so that they can be conveniently presented together at the end of the listing file. We include the real value of commodity consumption as discussed in Chapter 2, model M26, to show the difference between the effects of the reform on real commodity consumption (REALCONS) and true welfare (WELFARE), the latter accounting for changes in the value of leisure. Note from the results in the present case, that measuring only the change in real commodity consumption significantly overstates the true welfare gain of the tax reform (which is tiny) because of the fall in leisure (increase in labor supply).

Finally, although the difference is small, note that the capital tax that needs to be associated with TXL = 0 is TXK = 0.476, not the 0.50 value we assumed in the previous model adjusting the capital and labor taxes one for one in opposite directions.

\$TITLE Model M32: Closed 2x2 Economy -- Equal Yield Tax Reform \$ONTEXT

	Prod	uction S	Sectors		C	onsumers	
Markets	A	В	W	$ ext{TL}$	TK	CONS	
PX PY PW	120   	120	-120 -120 340			-340	
PLS PKS PL PK TAX	-48   -72   	-72 -48	-100	120 -100 -20	120 -100 -20	100 200 40	
\$OFFTEXT							
SETS S /1*5/;							
PARAMETERS TXL WELFARE(S) REALCONS(S) LABSUP(S) CAPTAX(S)	Real con Labor su	sumptior	n of good	S,			
\$ONTEXT							
\$MODEL:M32							
\$SECTORS: X Y W TL TK	! Acti ! Acti ! Supp	vity lev vity lev ly activ	vel for s vel for s vel for s vity for vity for	ector Y ector W L	(Hicksi	an welfare	e index)
\$COMMODITIES:  PX  PY  PL  PK  PLS  PKS	! Pric ! Pric ! Pric ! Pric	e index e index e index e index	for prim	odity Y ary fact ary fact ary fact	or K (note or L (g	et of tax) et of tax) ross of ta ross of ta	ıx)

```
PW
           ! Price index for welfare (expenditure function)
$CONSUMERS:
       CONS ! Income level for consumer CONS
$AUXILIARY:
               ! Endogenous capital tax from equal yield constraint.
       TXK
$PROD:X s:1
       O:PX Q:120
       I:PLS Q: 40 P:1.2
       I:PKS Q: 60 P:1.2
$PROD:Y s:1
       O:PY Q:120
       I:PLS Q: 60 P:1.2
       I:PKS Q: 40 P:1.2
$PROD:TL
       O:PLS Q:100 P:1.2
       I:PL Q:100 A:CONS T:TXL
$PROD:TK
       O:PKS Q:100 P:1.2
       I:PK Q:100 A:CONS N:TXK
$PROD:W s:0.7 a:1
       O:PW
             Q:340
       I:PX Q:120 a:
       I:PY Q:120 a:
       I:PL Q:100
$DEMAND: CONS
       D:PW Q:340
       E:PL Q:200
       E:PK Q:100
$CONSTRAINT:TXK
 TXL*PL*TL*100 + TXK*PK*TK*100 = E = 40 * (PX + PY)/2;
$OFFTEXT
$SYSINCLUDE mpsgeset M32
TXL = 0.20;
TXK.L = 0.20;
```

```
PLS.L = 1.2;
PKS.L = 1.2;
M32.ITERLIM = 0;
$INCLUDE M32.GEN
SOLVE M32 USING MCP;
M32.ITERLIM = 2000;
LOOP(S,
TXL = 0.25 - 0.05*ORD(S);
$INCLUDE M32.GEN
SOLVE M32 USING MCP;
WELFARE(S) = W.L;
REALCONS(S) = (PX.L*X.L*120 + PY.L*Y.L*120)
              /(PX.L**0.5*PY.L**0.5*240);
LABSUP(S) = TL.L;
CAPTAX(S) = TXK.L;
);
DISPLAY WELFARE, REALCONS, LABSUP, CAPTAX;
```

# Model 34 introduces a public good

The assumption of lump-sum redistribution is a convenient trick which simplifies tax policy analysis. In practice, governments often use money to purchase things which private markets do not provide. People value the public provision, but for some reason it is not easy to collect money from beneficiaries.

In this model, we first explicitly introduce government as an agent or "consumer" (GOVT). The tax revenue collected in the economy is assigned to the government. The government spends this on purchasing a good called G (price PG), which is produced from capital and labor like the other goods X and Y. Here is the production block for G (all inputs in all sectors are taxed at 25%) and the demand block for the consumer GOVT.

```
$PROD:G s:1
    O:PG    Q: 50
    I:PL    Q: 20    P:1.25    A:GOVT    T:TAX
    I:PK    Q: 20    P:1.25    A:GOVT    T:TAX
$DEMAND:GOVT    D:PG
```

The government is the only agent demanding PG in the model. As we noted back in Chapter 1, MPS/GE transfers the tax revenue to the government in the background, this being the government's only source of income.

Now comes the tricky part. Each consumer receives the full benefit of the public good without actually purchasing or paying for it. The community park is just there, free for the consumer. How do we model this free transfer from the government to the consumer?

To do this, we introduce another feature of MPS/GE, the rationing variable and rationing constraint. The rationing variable is declared as an auxiliary variable, denoted here LGP It appears as a multiplier on an endowment field in a consumer's demand equation. Here is the correct syntax for the demand blocks for the two consumers.

```
$DEMAND: CONS1
       D:PW1
               Q:125
               Q: 50
       E:PL
               Q: 50
       E:PK
       E:PG1
               Q: 50 R:LGP
$DEMAND: CONS2
       D:PW2
               Q:125
               Q: 50
       E:PL
       E:PK
               Q: 50
       E:PG2
               O: 50 R:LGP
```

MPS/GE just multiplies the quantity in the Q field by the value of the variable in the R field, so in this case CONS1 has an endowment of PG1 equal to 50\*LGP.

The value of LGP, like any auxiliary variable is set by a constraint equation. In our case, LGP takes on the value of G, the activity level for production of the public good. Note that 50 is the reference quantity for the G activity at G=1, so the auxiliary rationing variable LGP transfers the full amount of the public good to each consumer.

```
$CONSTRAINT:LGP
LGP =E= G;
```

Note that we have defined and used two different commodities for the public good endowments of the two consumers, PG1 and PG2 respectively, even though each consumer is getting the full (and therefore equal) amount of the public good. The reason that we do this is to prevent the consumers from being able to trade the goods in the event that they have different valuations of public good. One cannot sell his or her endowment of the public park to the other consumer. These "personalized" endowments of the public good are then inputs to each consumer's welfare. Here are the two utility functions.

```
$PROD:W1 s:1
       O:PW1
               Q:125
       I:PX
               Q: 70
       I:PY
               Q: 30
             Q: 50
       I:PG1
                      P:0.5
$PROD:W2 s:1
       O:PW2
               Q:125
               Q: 30
       I:PX
       I:PY
               Q: 70
       I:PG2
               Q: 50
                      P:0.5
```

Only consumer 1 is endowed wit PG1 and only consumer 1 demands PG1. Similar comments apply to consumer 2 and PG2.

What about the valuations of the public goods at a price PG1 = PG2 = 0.5? We are going to assume that the initial data represent an optimal initial provision of the public good. According to the Samuelson rule, discussed in connection with the next model, the public good is optimally provided with non-distortionary taxation if the cost of providing one unit equals the sum of the demand prices, since each consumer consumes the full amount. This is satisfied here if each consumer has a demand price of 0.5 in the initial benchmark situation.

Now we will give the full micro-consistent data matrix for this problem. Factor supplies are taxes at 25% to all sectors.

		Produc	ction	Secto	rs	C	Consumers		
Markets	X	Y	G	W1	W2	CONS1	CONS2	GOVT	
PX PY PG PL PK	-50 -30	100 -30	50 -20 -20	-70 -30	-30 -70	50 50	50 50	-50	
TAX	-20	-20	-10					50	
PW1 PW2 PG1 PG2				125 -25	125	-125 25	-125 25		_

The government is treated as another consumer, with an income balance condition that the GOVT column sum is zero. Taxes are treated like a market subject to a zero-row-sum, market-clearing condition: all tax revenues generated must be assigned to something. The personalized public goods are endowments of the consumers and inputs into consumers' welfare functions.

Take some time working through the rows and columns of this matrix. Note the zero row and column sums, and make sure that you understand each one.

Note finally that this data matrix is somewhat different from those used previous in that some of these entries are valuations are not observed in the market data (this was also true of the labor/leisure models, where the endowment of leisure and its valuation is not directly observed). The last four rows of the matrix are not observed in market data. We have constructed them under the assumptions that (a) the consumers have identical preferences for the public good, (b) the benchmark data represents a optimal provision of the public good.

Now we present the full model using these data. Note the specification of the initial values of variables not equal to one and the value of the auxiliary variable (default is zero, not 1) prior to the replication check (iterlim = 0).

\$TITLE Model M34: Closed 2x2 Economy -- Public Provision \$ONTEXT

	P	roduc	tion	Secto	rs	(	Consumers	
Markets	Х	Y	G	W1	W2	CONS1	CONS2	GOVT
PX   PY	100	100		-70 -30				
PG PL PK		-30 -50				50 50	50 50	-50
TAX	-20	-20	-10					50
PW1   PW2   PG1   PG2				125 -25	125	-125 25	-125 25	
\$OFFTEXT								

### **PARAMETERS**

TAX Tax rate on factor inputs to all sectors;

#### \$ONTEXT

## \$MODEL:M34

#### \$SECTORS:

X ! Activity level for sector X
Y ! Activity level for sector Y

G ! Activity level for sector G (public provision)

W1 ! Activity level for sector W1 (consumer 1 welfare index)
W2 ! Activity level for sector W2 (consumer 2 welfare index)

#### \$COMMODITIES:

 ${\tt PX} \quad ! \ {\tt Price index for commodity} \ {\tt X}$ 

PY ! Price index for commodity Y

PG ! Price index for commodity G(marginal cost of public output)

PL ! Price index for primary factor L (net of tax)

PK ! Price index for primary factor K

PW1 ! Price index for welfare (consumer 1)
PW2 ! Price index for welfare (consumer 2)

PG1 ! Private valuation of the public good (consumer 1)

```
PG2 ! Private valuation of the public good (consumer 2)
$CONSUMERS:
       CONS1
               ! Consumer 1
       CONS2
              ! Consumer 2
       GOVT
              ! Government
$AUXILIARY:
               ! Level of government provision
       LGP
$PROD:X s:1
       O:PX
             Q:100
       I:PL
               Q: 50 P:1.25 A:GOVT
                                      T:TAX
       I:PK
               Q: 30 P:1.25 A:GOVT
                                      T:TAX
$PROD:Y s:1
       O:PY
             Q:100
               Q: 30
       I:PL
                     P:1.25 A:GOVT
                                      T:TAX
       I:PK
               Q: 50
                      P:1.25 A:GOVT
                                      T:TAX
$PROD:G s:1
       O:PG
             Q: 50
               Q: 20
       I:PL
                      P:1.25 A:GOVT
                                     T:TAX
       I:PK
             Q: 20
                     P:1.25 A:GOVT T:TAX
$PROD:W1 s:1
       O:PW1
              Q:125
       I:PX
               Q: 70
       I:PY
               Q: 30
       I:PG1
              Q: 50
                     P:0.5
$PROD:W2 s:1
       O:PW2
               Q:125
       I:PX
               Q: 30
       I:PY
              Q: 70
               Q: 50
                     P:0.5
       I:PG2
$DEMAND:GOVT
       D:PG
$DEMAND: CONS1
       D:PW1
               Q:125
       E:PL
               Q: 50
       E:PK
              Q: 50
       E:PG1
              Q: 50
                     R:LGP
```

\$DEMAND: CONS2

```
D:PW2
                Q:125
        E:PL
                0: 50
        E:PK
                Q: 50
        E:PG2
                Q: 50 R:LGP
$CONSTRAINT:LGP
        LGP = E = G;
$OFFTEXT
$SYSINCLUDE mpsgeset M34
TAX = 0.25;
LGP.L = 1;
PG1.L = 0.5;
PG2.L = 0.5;
M34.ITERLIM = 0;
$INCLUDE M34.GEN
SOLVE M34 USING MCP;
M34.ITERLIM = 2000;
     The following counterfactuals check that the original
     benchmark is indeed an optimum by raising/lowering the tax
TAX = 0.20;
$INCLUDE M34.GEN
SOLVE M34 USING MCP;
TAX = 0.30;
$INCLUDE M34.GEN
SOLVE M34 USING MCP;
```

#### Exercises:

- (1) Define a set, parameters, and write a loop statement over finer values of the tax rate (e.g., 1% steps) storing the welfare of each consumer for each tax rate. See if the tax of 25% is optimal.
- (2) Change one consumer's valuation of the public good. The first run of the model should reproduce the benchmark data, but one consumer will prefer a different tax rate. Conduct the same experiment as in (1) with the new valuation and see what happens from the point of view of each consumer.

# Model 35 public good provision using the samuelson rule

This model is exactly the same as the previous one, except that the tax used to finance the public good is endogenous. Thus we will not repeat the data matrix here, and most other elements of the model are already familiar.

Instead of TAX being a parameter, it is now an auxiliary variable. Its value is set by the constraint equation:

```
$CONSTRAINT:TAX
PG =E= PG1 + PG2;
```

This is a condition noted by Samuelson many years ago. Since each consumer gets the full amount of the public good (the good is "non-rivaled"), the marginal benefit of another unit of the good is the sum of the demand prices for all the consumers. Efficiency is achieved when this sum of benefits is equal to the marginal cost of producing another unit. This is given by the above equation.

Note that the auxiliary variable itself need not appear in the constraint equation associated with it. The solution algorithm will adjust TAX in order to satisfy this condition.

Strictly speaking, the Samuelson rule is valid only if the tax needed to pay for the public good can be raised in a non-distortionary way. If distortionary taxes must be used, the sum of marginal benefits must be weighed against the marginal cost of production plus the marginal burden of taxation. It is beyond the scope of this chapter to discuss this problem further here.

When we run this model, we will get back a value of TAX = 0.25, because we calibrated the prefences assuming that the initial data was optimal. As a counterfactual experiment, we change one consumer's valuation of the public good, using a parameter VG1 which is a multiplier on consumer 1's initial valuation of the good.

```
$PROD:W1 s:1
O:PW1 Q:125
I:PX Q: 70
I:PY Q: 30
I:PG1 Q: 50 P:(0.5*VG1)
```

In the counterfactual experiment we double consumer 1's "willingness to pay", setting VG1 = 2. Here are some results from the counterfactual.

 VAR	X	•	0.909	+INF	
 VAR	Y	•	0.909	+INF	
 VAR	G	•	1.364	+INF	
 VAR	W1	•	1.041	+INF	
 VAR	W2	•	0.986	+INF	
 VAR	TAX	•	0.375	+INF	

With consumer 1's increased valuation of the public good, it is optimal to raise the tax from 0.25 to 0.375 as shown. Resources are transferred out of producing final goods X and Y and into producing G. The activity for G rises from 1.0 in the benchmark to 1.364.

Note that, although the high tax is efficient according to the Samuelson rule, it nevertheless results in a redistribution of welfare from the low valuation consumer to the high valuation consumer.

There are many uses for the type of modeling in public and environmental economics. Environmental economists in particular devote large amounts of effort to soliciting individuals' preferences for non-market goods. The results of such surveys and studies can serve as inputs into calibrating the preferences for models such as this one. When calibrated, it is unlikely that the initial level of public goods (or environmental quality) is optimal. The following program can then be used to find what the optimum level and optimal taxes are.

\$TITLE Model M35: Closed 2x2 Economy - Public Output with Samuelson Rule This model is the same as M34 except that the tax to finance the public good is set endogenously PARAMETER Preference index for public goods for consumer 1; VG1 = 1;\$ONTEXT \$MODEL:M35 \$SECTORS: Χ ! Activity level for sector X ! Activity level for sector Y Y ! Activity level for sector G (public provision) G ! Activity level for sector W1 (consumer 1 welfare index) W1! Activity level for sector W2 (consumer 2 welfare index) W2\$COMMODITIES: ! Price index for commodity X PX! Price index for commodity Y PG ! Price index for commodity G (marg cost of public output) ! Price index for primary factor L (net of tax) PL! Price index for primary factor K PΚ ! Price index for welfare (consumer 1) PW1 ! Price index for welfare (consumer 2) PW2 ! Private valuation of the public good (consumer 1) PG1 ! Private valuation of the public good (consumer 2) PG2 \$CONSUMERS: ! Consumer 1 CONS1 ! Consumer 2 CONS2 GOVT ! Government. \$AUXILIARY: LGP ! Level of government provision ! Uniform value-added tax rate TAX \$PROD:X s:1

O:PX

I:PL I:PK Q:100

Q: 50 P:1.25 A:GOVT N:TAX

Q: 30 P:1.25 A:GOVT N:TAX

```
$PROD:Y s:1
        O:PY
                Q:100
             Q: 30 P:1.25 A:GOVT N:TAX
Q: 50 P:1.25 A:GOVT N:TAX
        I: PL
        I:PK
$PROD:G s:1
        O:PG
             Q: 50
             Q: 20 P:1.25 A:GOVT N:TAX
Q: 20 P:1.25 A:GOVT N:TAX
        I: PL
        I:PK
$PROD:W1 s:1
        O:PW1
                Q:125
        I:PX
               Q: 70
        I:PY
               Q: 30
        I:PG1 Q:(VG1*50) P:0.5
$PROD:W2 s:1
        O:PW2
              Q:125
        I:PX
               Q: 30
               Q: 70
        I: PY
        I:PG2 Q: 50 P:0.5
$DEMAND:GOVT
        D:PG
$DEMAND: CONS1
        D:PW1 Q:125
        E:PL
               Q: 50
        E:PK
               Q: 50
        E:PG1 Q: 50 R:LGP
$DEMAND: CONS2
        D:PW2 Q:125
                Q: 50
        E: PL
                Q: 50
        E:PK
        E:PG2
                Q: 50
                       R:LGP
$CONSTRAINT:LGP
        LGP = E = G;
$CONSTRAINT: TAX
        PG = E = PG1 + PG2;
$OFFTEXT
$SYSINCLUDE mpsgeset M35
```

\* Benchmark replication

```
TAX.L = 0.25;
LGP.L = 1;
PG1.L = 0.5;
PG2.L = 0.5;

M35.ITERLIM = 0;
$INCLUDE M35.GEN
SOLVE M35 USING MCP;
M35.ITERLIM = 2000;

* What happens to consumer 2 welfare if consumer 1 decides
* she would like more public output:

VG1 = 2;
$INCLUDE M35.GEN
SOLVE M35 USING MCP;
```

## Exercise:

Run this counterfactual again, except fix the tax rate at its initial optimal value: TAX.FX = 0.25. See how this changes the welfare of the two consumers.

# Model 36 taxes and classical unemployment

In many situations, initial data or the questions that a modeler may wish to analyze do not correspond to an equilibrium with all markets clearing. Unemployment is an obvious situation. While MPS/GE is designed to compute equilibria with markets clearing, there are some simple tricks to incorporate a rich set of situations and policy options into a model.

This model considers "classical" unemployment, by which we will simply mean that there is a wage which is rigid downward, perhaps a legal minimum wage. This constraint may or may not be binding in equilibrium, and the complementarity features of GAMS and MPS/GE make this easy to compute. Other situations that the technique applies to include the analysis of quotas, which we will examine in the next chapter.

The model we use here is essentially the same as the one we used in M21, where we considered tax reform as a counterfactual experiment. The data matrix shown below is the same one used in M21, except here we specify the total endowment of labor as 100 units, but an initial unemployment rate of 20% limits the amount of labor used in X and Y to 80 units. The parameter U0 is thus set at 0.20 initially. As was the case in M21, we assume that there is initially a tax of 100% on the labor used in the production of X. The consumer (supply) price of labor is chosen to be one, so the producer price (user cost) is 2, and this must appear in the price field for the X production block.

We model unemployment with the use of an auxiliary variable and a constraint equation. The auxiliary variable is the unemployment rate, and is a rationing multiplier on the consumer's endowment of labor.

```
$DEMAND:CONS

D:PW Q:200

E:PL Q:(80/(1-U0))

E:PL Q:(-80/(1-U0)) R:U

E:PK Q:100
```

MPS/GE does not permit an arithmetic operation in an R field, so we specify a second row giving an negative endowment of labor, which will work unless the overall endowment becomes negative. Incidentally, any arithmetic operation in a Q or P field must be enclosed in parentheses. We include U0 in the Q field here rather than just put 100 so that the modeler can change this parameter if desired. Note that at the initial unemployment rate of 20% the values of the two E fields sum to 80, the amount used in production in the benchmark.

In setting the minimum wage constraint, the modeler must decide on what defines the minimum level. Minimum in terms of what? Here we are going to chose PW, the price of buying one unit of utility, or in other words the real consumer price index. If PW is chosen as the numeraire which we have been doing in these exercises, then the nominal value of PL is fixed, but this will not be the case if something else is chosen as numeraire. In any case, the

point is that the modeler should make a conscious choice of the numeraire for the minimum wage rate. In many countries it is periodically adjusted to reflect changes in the consumer price index, and PW is of course the theoretically ideal consumer price index.

Since units are chosen such that the prices PL and PW are initially equal to one, the constraint equation setting the value of the auxiliary variable U is given by:

```
$CONSTRAINT:U
PL =G= PW;
```

Running the model verifies that we have benchmarked it correctly, and that the initial unemployment rate is 20%.

For a counterfactual experiment, we introduce a tax reform as in Model 21, and replace the 100% tax on labor in X with equal tax rates of 25% on both factors (although this is still a distortionary tax. Now run the model. Some of the results are as follows:

		LOWER	LEVEL	UPPER	MARGINAL
 VAR	Х		1.178	+INF	
 VAR	Y	•	1.106	+INF	•
 VAR	W		1.142	+INF	
 VAR	PX	•	0.969	+INF	•
 VAR	PY		1.032	+INF	
 VAR	PL	•	1.051	+INF	•
 VAR	PK		1.005	+INF	
 VAR	PW	1.000	1.000	1.000	EPS
 VAR	CONS		228.374	+INF	
 VAR	U	•	•	+INF	0.051

The minimum wage ceases to be binding (PL = 1.051) and unemployment falls to zero (note that the marginal on U is the excess of PL over PW). There is a very large increase in welfare, which goes to 1.142 as the increases in employment reinforces the tax reform. There are some very interesting policy economics in this example. Initially, we have two distortions that are reinforcing one another. If we reform the tax structure, the second distortion is automatically removed by making it non-binding.

To see this effect, we do one final counterfactual in the model, which is to fix the unemployment rate at its initial value of 20%, but allow the tax reform. If you look at the listing file, you will see that welfare rises to only 1.021. Most of the effect of the tax reform shown above is its indirect effect in eliminating the binding minimum wage constraint.

\$TITLE Model M36: Closed 2x2 Economy - Taxes and Classical Unemployment \$ONTEXT

	Production Sectors			Consumers	
Markets	X	Y	W	CONS	
PX PY PW PL PK TAX	-20 -60 -20	100 -60 -40 0	-100 -100 200	-200   100*(1-U)   100   20	

### \$OFFTEXT

### **PARAMETERS**

Proportional output tax on sector X, TYProportional output tax on sector Y, Ad-valorem tax on labor inputs to X, TLXAd-valorem tax on capital inputs to X, TKX U0 Initial unemployment rate;

U0 = 0.20;

### \$ONTEXT

\$MODEL:M36

#### \$SECTORS:

! Activity level for sector X ! Activity level for sector Y Y

! Activity level for sector W (Hicksian welfare index)

## \$COMMODITIES:

PΧ ! Price index for commodity X ! Price index for commodity Y

! Price index for primary factor L (net of tax) PL

! Price index for primary factor K PΚ

PW! Price index for welfare (expenditure function)

#### \$CONSUMERS:

CONS ! Income level for consumer CONS

#### \$AUXILIARY:

U ! Unemployment rate

```
$PROD:X s:1
        O:PX
              Q:100 A:CONS T:TX
        I:PL Q: 20 P:2 A:CONS T:TLX I:PK Q: 60 A:CONS T:TKX
$PROD:Y s:1
        O:PY
                Q:100 T:TY
              Q: 60
        I:PL
        I:PK
              Q: 40
$PROD:W s:1
        O:PW
                Q:200
        I:PX
              Q:100
        I:PY
              Q:100
$DEMAND: CONS
        D:PW
             Q:200
        E:PL Q:(80/(1-U0))
       E:PL
              Q:(-80/(1-U0)) R:U
                Q:100
$CONSTRAINT:U
     PL = G = PW;
$OFFTEXT
$SYSINCLUDE mpsgeset M36
PW.FX = 1;
TX = 0;
TY = 0;
TLX = 1;
TKX = 0;
U.L = U0;
M36.ITERLIM = 0;
$INCLUDE M36.GEN
SOLVE M36 USING MCP;
M36.ITERLIM = 2000;
        As in M31, we replace the tax on labor inputs
        by a uniform tax on both factors:
TLX = 0.25;
TKX = 0.25;
TX = 0;
TY = 0;
```

\$INCLUDE M36.GEN
SOLVE M36 USING MCP;

U.FX = 0.20;

\$INCLUDE M36.GEN
SOLVE M36 USING MCP;

# Exercises:

- (1) Try a different numeraire for defining the minimum wage and see what happens.
- (2) If you want to be a little bit more ambitious, try adding a labor-leisure decision to the problem.

# Model 37 steady-state capital stock

As in the case of labor supply, we would like to have models in which the stock of capital is endogenous. This requires dynamic modeling, a significant complication of what we have done to this point. This chapter will not consider true dynamic modeling. However, we will present a shortcut that is very valuable in many situations. In models for which there exists a steady state, it is possible to represent this steady state as a complementarity problem not that different from our static models. We can then at least perform comparative steady-state experiments, in which a parameter change moves us from one steady state to another. But actually solving for the transition path requires a more complicated analysis that is postponed until later in the book.

Model 37 incorporates optimal capital accumulation via the use of a rationing constraint and endogenous "taxes" to create a model for comparative steady state analysis when the capital stock is endogenously adjusted to its steady-state value. The labor force is assumed fixed in this model.

Let r denote the rental price (for one period) of a unit of capital and let  $p_k$  denote the price of a new unit of capital.  $\delta$  will denote the rate of depreciation of capital per period, and  $\rho$  will denote the discount rate between periods. The steady-state optimal capital accumulation condition is a relationship between the price and rental rate on capital. The rental rate must be given by:

$$r = \left[1 - \frac{1 - \delta}{1 + \rho}\right] p_k$$

The rental price (r) is equal to the price of creating a new unit of capital  $(p_k)$  minus the present value of what the capital could be sold for next period. In the steady state, the price of a new unit next period is the same, so an old unit can be sold for its original value minus one period's depreciation  $(1 - \delta)$ . The present value of the undepreciated portion is thus  $(1 - \delta)/(1 + \rho)$ .

The dynamic steady-state problem is represented as a static problem using two tricks. First, the use of capital is subsidized to create the desired wedge (denoted TAU) between the rental price and the price of new capital. Following the declaration of the parameters DELTA, RHO, and TAU you will see:

$$TAU = - (1 - DELTA) / (1 + RHO);$$

This then appears as a subsidy to capital use, creating the wedge between the rental price and the price of producing a new unit of capital (PK).

```
I:PK Q:120 P:0.5 A:CONS T:TAU

$PROD:Y s:1
    O:PY Q:100
    I:PL Q: 60
    I:PK Q: 80 P:0.5 A:CONS T:TAU
```

One unit of new capital is produced using one unit of labor, as you will see in the production block for activity K.

Second, since in the steady state newly produced capital is equal to depreciation and depreciation is equal to a share DELTA of total capital, a rationing constraint is specified to "endow" consumers with the carryforward from the previous period. Let  $K_s$  denote the capital stock and  $K_n$  the production of new capital. In the steady state, these are related by:

$$K_n = \delta K_s \qquad K_s = \frac{K_n}{1-\delta}$$

The carryforward from the previous period is the steady state stock minus new production. It is is called KFORWD and given by:

carry forward = 
$$KFORWD = K_s - K_n = \frac{K_n}{\delta} - K_n = \frac{(1 - \delta)K_n}{\delta}$$

So in the model below, we will give the consumer an endowment, via the rationing multiplier KFORWD, the quantity of capital on the right-hand side of the above equation.

```
$DEMAND:CONS
D:PW
E:PL Q:160
E:PK Q:1 R:KFORWRD

$CONSTRAINT:KFORWRD
KFORWRD =E= K * (1-DELTA) / DELTA;
```

That completes the model description. As has been our practice, we include the initial micro-consistent data matrix in the program. Study this carefully in going through the program. Note that in the data matrix, the consumer is "endowed" with 140 units of capital. In the program, this is the carry forward and is subject to change in counterfactual experiments.

Two counterfactuals are run in the program. First the rate of time preference  $\rho$  is raised. In the second,  $\rho$  is set back to its initial value, and there is a tax on new capital production. See if you can guess what the effects of these changes should be, and then look at the results.

\$TITLE Model M37: Closed 2x2 Economy - Steady State Capital Stock \$ONTEXT

	D 1					
	Production	n Sectors	Consumers			
Markets	X Y	K W	CONS			
PX PY PW PL PK SUB	100   100   -40 -60   -120 -80   60 40	200 -60	   -200   160   140   -100			
\$OFFTEXT						
PARAMETERS  RHO Time preference parameter,  DELTA Depreciation rate,  TAU Effective capital use tax (it is a subsidy, its < 0),  KTAX Tax on new capital production  NEWCAP New capital stock after counterfactual (= 1 initially);						
RHO = 0.4; DELTA = 0.3; TAU = - (1 - DELTA) / (1 + RHO); KTAX = 0;						
\$ONTEXT						
\$MODEL:M37						
\$SECTORS:  X Y W K	! Activity			index)		
\$COMMODITIES: PX PY PL PK PW	! Price inde ! Price inde ! Price inde	ex for primary	ity Y y factor L (net of tax)	)		

```
$CONSUMERS:
       CONS ! Income level for consumer CONS
$AUXILIARY:
       KFORWRD ! Capital stock from previous period
$PROD:X s:1
       O:PX Q:100
       I:PL Q: 40
       I:PK Q:120 P:0.5 A:CONS T:TAU
$PROD:Y s:1
       O:PY Q:100
       I:PL Q: 60
       I:PK Q: 80 P:0.5 A:CONS T:TAU
$PROD:K
       O:PK Q:1 A:CONS T:KTAX
       I:PL Q:1
$PROD:W s:1
       O:PW Q:200
       I:PX Q:100
       I:PY Q:100
$DEMAND: CONS
       D:PW
       E:PL Q:160
       E:PK Q:1 R:KFORWRD
$CONSTRAINT: KFORWRD
       KFORWRD =E= K * (1-DELTA) / DELTA;
$OFFTEXT
$SYSINCLUDE mpsgeset M37
K.L = 60;
KFORWRD.L = 140;
PW.FX = 1;
$INCLUDE M37.GEN
SOLVE M37 USING MCP;
```

```
* Raise the rate of time preference from 0.4 to 0.6:

RHO = 0.6;

TAU = - (1 - DELTA) / (1 + RHO);

$INCLUDE M37.GEN
SOLVE M37 USING MCP;

NEWCAP = K.L/60;
DISPLAY NEWCAP;

* Set rho back to 0.4, tax new capital at 0.20

RHO = 0.4;

TAU = - (1 - DELTA) / (1 + RHO);

KTAX = 0.20;

$INCLUDE M37.GEN
SOLVE M37 USING MCP;

NEWCAP = K.L/60;
DISPLAY NEWCAP;
```

### Exercises:

- (1) Make sure you understand the economic intuition behind the results in the counterfactuals.
- (2) If you are ambitious, write a new model combining endogenous labor supply and an equal yield tax constraint with this model. Do some experiments to see if the capital tax or the labor tax is more distortionary. This is an important use of applied modeling: theory can never tell you which of two distortionary taxes is more distortionary.